Ribbon format cb0_1_1_1_1_1
c is the input counter
b is the input byte
0 is the delimiter between input byte and binary format value
_s are filled with a digit in the binary format value in the reverse order
1s are for easy navigation between input byte and binary format value
>> >>+>>+>>+>>+>>+							Set in binary format indicators
<<<<< <<<<< <<								Get back to the initial position
+++++ +++++ +++++ +++++ +++++ + [			Input counter for 26 characters
	>,										Scan the byte
	----- ----- ----- ----- -----
	----- ----- ----- ----- -----
	----- ----- ----- ----- -----
	----- ----- ----- ----- -- [			Decrease by ascii value of 'a'
		>> [
			->>
		]
		+<									Increase the binary format value
		[
			<<
		]
		<-									Go back to the byte and decrement
	]
	>>>>> >>>>> > [							Go to binary format end
		< [									Print *
			+++++ +++++ +++++ +++++ +++++
			+++++ +++++ +++++ +.
			----- ----- ----- ----- -----
			----- ----- ----- -->-<
		] > [								Print hyphen
			-<
			+++++ +++++ +++++ +++++ +++++
			+++++ +++++ +++++ +++++ .
			----- ----- ----- ----- -----
			----- ----- ----- ----->
		]
		+<<
	]										Print binary format in the reverse
	+++++ +++++.							Print new line
	----- -----
	<<-										Go to the input counter and decrement
]
